*Eintritte in Alo werden mit zukünftigen Outcomes verknüpft

capture log close
log using $log/AED_dataprep_09_outcomes, text replace

clear
clear matrix
set maxvar 10000
set matsize 800
set more off


set trace off
      
	  
/* ------------------------------------------------------------------------------------------------- */
/* --- Merge entries into unemployment with future outcomes ---------------------------------------- */
/* ------------------------------------------------------------------------------------------------- */



foreach iebversion in v901 {
	foreach out in monthly {
		
		if "`out'"=="monthly" {
		local help_out ""
		local help_mon "_monthly"
		}
		
		forvalue year = 2001/2004 {
		
				di "*****************************************************"
				di "Year: `year'"
				di c(current_time)
				di "*****************************************************"
   			
				/*Restrict time window to 2 years starting at entry into unemployment*/
				local yearafter = `year'+2
				local what "`year'/`yearafter'"
	           		
			
				forvalue jahr = `what' {
				
				use $data/eintritte_ue_`year'_`iebversion', clear
				keep iza_id iza_eintrittm iza_eintritty
				
				joinby iza_id using $data/outcome`help_mon'_y`jahr'_`iebversion', unmatched(master)
				tab _merge
				drop _merge
	
				joinby iza_id using $data/outcome_mth_y`jahr'_`iebversion', unmatched(master)
				tab _merge
				drop _merge
				
				gen eintritt = .
				tostring eintritt, replace
				gen help = iza_eintrittm
				tostring iza_eintritty, replace
				tostring iza_eintrittm, replace
				replace eintritt = substr(iza_eintritty,1,4) + "0" + substr(iza_eintrittm,1,2) if help < 10
				replace eintritt = substr(iza_eintritty,1,4) + substr(iza_eintrittm,1,2) if help >=10
				destring eintritt, replace
				destring iza_eintritty, replace
				
				
				keep iza_id iza_eintritty iza_eintrittm eintritt iza_beh_* iza_leh_* iza_asu_* iza_lhg_* iza_geld_* iza_mth1_* iza_mth2_* iza_mth3_* iza_mth4_* iza_azeit_* iza_bet_*
				
				
				reshape long iza_beh_ iza_leh_ iza_asu_ iza_geld_beh_ iza_geld_leh_ iza_mth1_ iza_mth2_ iza_mth3_ iza_mth4_ iza_bet_ iza_azeit_, i(iza_id) j(lfd)
		
				sort iza_id lfd
				
				saveold $data/outcome_`year'_`jahr',replace
				
				}/*what*/
				
				
				use $data/outcome_`year'_`year',clear
                
                local yearafter = `year'+1
               local yearend = `year'+2
	           local what "`yearafter'/`yearend'"
	           
	           	
                
                forvalues jahr = `what' {
                append using $data/outcome_`year'_`jahr'
				}
				sort iza_id lfd
				
				
				/*Keep period of two years starting at entry into unemployment*/
				by iza_id: drop if lfd < eintritt
				by iza_id: gen end = eintritt+200
				by iza_id: drop if lfd >= end
				drop end
				
				
				/*Redefine entry month if unemployment just started in the last week of the initial entry month*/
				
				by iza_id: gen ue_ = 1 if (iza_asu_==31 | iza_asu_==32 | (iza_asu_==33 & iza_beh_==.)) & iza_mth1_==.
				by iza_id: gen help = 1 if ue_[1] ~= 1 & ue_[2]  == 1
				by iza_id: replace iza_asu_ = 31 if _n==1 & help==1
				drop ue_ help
				
				
				gen sv_ = 0
				gen ue_ = 0
				
				sort iza_id lfd
	
				#delimit;

				/*Unemployment with ALMP*/
				by iza_id: replace ue_ = 1 if (iza_asu_==31 | iza_asu_==32 | (iza_asu_==33 & iza_beh_==.) | iza_leh_~=.
	 			       | iza_mth1_~=.) 
				       ;

				/*Regular Employment*/
				by iza_id: replace sv_ = 1 if (iza_beh_==101 |
					iza_beh_==112 | 
					iza_beh_==120 | 
					iza_beh_==140 |
					iza_beh_==143 | 
					iza_beh_==203 
					)
	                                 & ue_~=1;
				
				#delimit cr
				
		
		/*Daily earnings from employment*/
		gen geldsv_= 0
		by iza_id: replace geldsv_ = iza_geld_beh_ if sv_ == 1
		
		
		/*transfer payments*/
		gen geldleh_= 0
		by iza_id: replace geldleh_ = iza_geld_leh_ if ue_ ==1
		
		/*Drop individuals with unemployment duration < 2 weeks*/
		by iza_id: gen help = 1 if ue_[1]~=1 
		codebook iza_id
		drop if help == 1  
		drop help
		codebook iza_id
		
		keep iza_id lfd iza_eintritty iza_eintrittm eintritt sv_ ue_ geldsv_ geldleh_ iza_beh_ iza_asu_ iza_leh_ iza_mth1_ iza_mth2_ iza_mth3_ iza_mth4_ iza_bet_ iza_azeit_
		
		
		
	/*Generate ALMP categories*/
	
	forvalues x =1/4 {
	
	#delimit;
	replace iza_mth`x'_ = 1  if (iza_mth`x'_ >=10100 & iza_mth`x'_<=10299)| iza_mth`x'_ == 13000| (iza_mth`x'_ >= 25001 & iza_mth`x'_ <=25099);
	replace iza_mth`x'_ = 2  if (iza_mth`x'_ >=10501 & iza_mth`x'_<=10999)| (iza_mth`x'_ >=11301 & iza_mth`x'_<=11499)|  
				    iza_mth`x'_==13001 | iza_mth`x'_ == 15990| iza_mth`x'_ == 23001;
	replace iza_mth`x'_ = 3  if  iza_mth`x'_==11101 | iza_mth`x'_ ==11199 | iza_mth`x'_==11801 | iza_mth`x'_==11899 | 
				   (iza_mth`x'_ >= 12001 & iza_mth`x'_ <= 12099)|  
				    iza_mth`x'_==13003| iza_mth`x'_ ==14117 | iza_mth`x'_ ==14127| iza_mth`x'_==15590|
				    iza_mth`x'_==15690 | iza_mth`x'_ == 23003| (iza_mth`x'_ >=24001& iza_mth`x'_<=24099);
   	replace iza_mth`x'_ = 4  if iza_mth`x'_ == 13007| iza_mth`x'_==13807| iza_mth`x'_==13837|iza_mth`x'_==13847| (iza_mth`x'_>=16021 & iza_mth`x'_<=16023) | iza_mth`x'_ ==16025|
				    iza_mth`x'_ == 23007;
	replace iza_mth`x'_ = 5  if (iza_mth`x'_ >=14002 & iza_mth`x'_<=14020) | (iza_mth`x'_ >=15110 & iza_mth`x'_<=15120)|
				    iza_mth`x'_ == 23005;				    
	replace iza_mth`x'_ = 6  if iza_mth`x'_ == 14001 | (iza_mth`x'_ >=14031 & iza_mth`x'_<=14034) | (iza_mth`x'_ >=15131 & iza_mth`x'_<=15134)| 
				   (iza_mth`x'_ >=15432 & iza_mth`x'_<=15434) ;
	replace iza_mth`x'_ = 7  if (iza_mth`x'_ >=14040 & iza_mth`x'_<=14041) | (iza_mth`x'_ >=15140 & iza_mth`x'_<=15141)|
				    iza_mth`x'_==15241| (iza_mth`x'_ >=15440 & iza_mth`x'_<=15441) | iza_mth`x'_==13002|
				    iza_mth`x'_== 23002;
	replace iza_mth`x'_ = 8  if (iza_mth`x'_ >=14050 & iza_mth`x'_<=14099)| (iza_mth`x'_ >=15150 & iza_mth`x'_<=15170)|
				    iza_mth`x'_ == 15199| iza_mth`x'_==15460; 
	replace iza_mth`x'_ = 9 if iza_mth`x'_ ==14112 | iza_mth`x'_ ==14113 |
				   iza_mth`x'_ ==14115| iza_mth`x'_ ==14116 | iza_mth`x'_ ==14122 | iza_mth`x'_ ==14123 |
				   iza_mth`x'_ ==14125| iza_mth`x'_ ==14126 | iza_mth`x'_ ==14132 | iza_mth`x'_ ==14133 |
				   iza_mth`x'_ ==14135| iza_mth`x'_ ==14192 | iza_mth`x'_ ==14193 | iza_mth`x'_ == 15302 |
				   iza_mth`x'_ ==15303 | iza_mth`x'_ ==13009 | iza_mth`x'_ ==14104 | iza_mth`x'_ == 14114 |iza_mth`x'_ ==14124 ;
	replace iza_mth`x'_ = 10 if iza_mth`x'_ ==14134 | iza_mth`x'_ ==14194 | iza_mth`x'_ == 14280| iza_mth`x'_ == 15304 | iza_mth`x'_ ==14101 | iza_mth`x'_ ==14111 | iza_mth`x'_ ==14119 | iza_mth`x'_ ==14121;
	replace iza_mth`x'_ = 11 if iza_mth`x'_ ==14129 | iza_mth`x'_ ==14131 | iza_mth`x'_ ==14139 | iza_mth`x'_ ==14191| 
				   iza_mth`x'_ ==14199 |  iza_mth`x'_==15301 | iza_mth`x'_==15399 ;	
	replace iza_mth`x'_ = 12 if iza_mth`x'_ ==14280 | iza_mth`x'_ == 15180| (iza_mth`x'_ >=15480 & iza_mth`x'_<=15490);
	replace iza_mth`x'_ = 13 if iza_mth`x'_ ==16004  | iza_mth`x'_==16071;
	replace iza_mth`x'_ = 14 if iza_mth`x'_ ==16003 |(iza_mth`x'_ >=16005 & iza_mth`x'_<=16006) ; 
	replace iza_mth`x'_ = 15 if iza_mth`x'_ ==16008 |(iza_mth`x'_ >=16087 & iza_mth`x'_<=16088)| iza_mth`x'_ == 19001 ;
	replace iza_mth`x'_ = 16 if iza_mth`x'_ ==16009 |(iza_mth`x'_ >=19002 & iza_mth`x'_<=19005)| (iza_mth`x'_>= 20001 & iza_mth`x'_ <=20002);
	replace iza_mth`x'_ = 17 if iza_mth`x'_==16072 | iza_mth`x'_==16073; 
	replace iza_mth`x'_ = 18 if iza_mth`x'_ ==16074;
	replace iza_mth`x'_ = 19 if (iza_mth`x'_ >=16010 & iza_mth`x'_<=16011) | iza_mth`x'_ ==16024 | iza_mth`x'_ ==13005| 
				   iza_mth`x'_==13008;
	replace iza_mth`x'_ = 20 if (iza_mth`x'_ >= 27000 & iza_mth`x'_ <=27027)|(iza_mth`x'_ >= 27069 & iza_mth`x'_ <=27088) ;
	replace iza_mth`x'_ = 21 if (iza_mth`x'_ >= 27038 & iza_mth`x'_ <=27050) ;
	replace iza_mth`x'_ = 22 if iza_mth`x'_ ==13006 | (iza_mth`x'_ >= 11701 & iza_mth`x'_ <=11799) | iza_mth`x'_ ==16086|
				  (iza_mth`x'_ >= 21001 & iza_mth`x'_ <=21099) ;
	replace iza_mth`x'_ = 23 if iza_mth`x'_ ==13004|iza_mth`x'_ == 23004 ;
	replace iza_mth`x'_ = 24 if (iza_mth`x'_ >= 11501 & iza_mth`x'_ <=11599);
	replace iza_mth`x'_ = 25 if (iza_mth`x'_ >=10301 & iza_mth`x'_<=10313) | (iza_mth`x'_ >=10315 & iza_mth`x'_<=10499);  
	replace iza_mth`x'_ = 26 if (iza_mth`x'_ == 10314);
	replace iza_mth`x'_ = 27 if iza_mth`x'_==15499| iza_mth`x'_ == 15890| iza_mth`x'_ == 15999| iza_mth`x'_== 16099|
				   (iza_mth`x'_ >=11601 & iza_mth`x'_ <=11602) | (iza_mth`x'_>=22003 & iza_mth`x'_ <=22005) |
				   (iza_mth`x'_ >=11901 & iza_mth`x'_ <=11902) | iza_mth`x'_ == 16033 | (iza_mth`x'_ >=23008 & iza_mth`x'_ <=23009);
	;
	#delimit cr
	
	
	}
	
	
	
	*Any ALMP participation
	gen iza_almp_=1 if iza_mth1_>=1 & iza_mth1_<=27
	
	*Longterm training: Kat 5,6,7 8
	gen iza_ltt_=1 if (iza_mth1_==5|iza_mth1_==6|iza_mth1_==7|iza_mth1_==8)
	replace iza_ltt_=1 if (iza_mth2_==5|iza_mth2_==6|iza_mth2_==7|iza_mth2_==8)		
	replace iza_ltt_=1 if (iza_mth3_==5|iza_mth3_==6|iza_mth3_==7|iza_mth3_==8)
	replace iza_ltt_=1 if (iza_mth4_==5|iza_mth4_==6|iza_mth4_==7|iza_mth4_==8)
	
	
	*Wage subsidy: Kat 2
	gen iza_ws_=1 if (iza_mth1_==2|iza_mth2_==2|iza_mth3_==2|iza_mth4_==2)
	
	*short-term training: Kat9, 10, 11
	gen iza_stt_=1 if (iza_mth1_==9|iza_mth1_==10|iza_mth1_==11)
	replace iza_stt_=1 if (iza_mth2_==9|iza_mth2_==10|iza_mth2_==11)		
	replace iza_stt_=1 if (iza_mth3_==9|iza_mth3_==10|iza_mth3_==11)
	replace iza_stt_=1 if (iza_mth4_==9|iza_mth4_==10|iza_mth4_==11)
	
	*Workfare: Kat 1
	gen iza_workfare_=1 if (iza_mth1_==1|iza_mth2_==1|iza_mth3_==1|iza_mth4_==1)
		
		
		
		
		compress				
		saveold  $data/entries_outcome_`year'_`iebversion', replace
			
				*Delete temporary files
				local yearend = `year'+2
				local what "`year'/`yearend'"
				forvalues jahr = `what' {
                erase $data/outcome_`year'_`jahr'.dta
				}
			
			}/* year*/
			
		}/* out*/
		
		}
	
	
log close
			
				
				
